

Journal of Statistical Computation and Simulation
Publication details, including instructions for authors and subscription information:
http://www.tandfonline.com/loi/gscs20

Computation of the highest posterior density interval in bayesian analysis

Noyan Turkkan [a] & T. Pham-Gia [b]

[a] School of Engineering , Université de Monoton , Monoton, N B, ElA 3E9, Canada

[b] Faculty of Science , Université de Monoton , N B, ElA 3E9, Canada

Published online: 20 Mar 2007.

J. Statist. Comput. Simul., Vol. 44, pp. 243–250
Reprints available directly from the publisher
Photocopying permitted by license only

COMPUTATION OF THE HIGHEST POSTERIOR DENSITY INTERVAL IN BAYESIAN ANALYSIS

NOYAN TURKKAN

School of Engineering, Université de Moncton, Moncton, N.B., E1A 3E9 Canada

T. PHAM-GIA

Faculty of Science, Université de Moncton, Moncton, N.B., E1A 3E9 Canada

(Received January 24, 1992; in final form July 1, 1992)

KEY WORDS: Posterior density, Bayes, Credible interval.

DESCRIPTION AND PURPOSE

Subroutine HPD computes the $(1 - \alpha)$-highest posterior density interval for an input posterior density and a given value of α. The output is either a finite interval, a combination of disjoint intervals, a left bounded or a right bounded interval.

THEORY

In univariate Bayesian analysis, all inferences on the parameter θ may be made from the posterior density $f(\theta|\vec{x})$, where $\vec{x}$ is the observed sample. For α given, different intervals with probability $(1 - \alpha)$ can be computed, using $f(\theta|\vec{x})$, and are called $(1 - \alpha)$-credible intervals, or Bayesian confidence intervals (although originally, these names were used only when the priors are non-informative). Among these intervals, the highest posterior density interval (or HPD interval) is the one having the additional property that every point within it has a higher probability than any point without it. It is also the shortest $(1 - \alpha)$-credible interval (Box and Tiao (1973, p. 85)). Hence, a HPD interval $I_{1-\alpha}$ is the complement of the set I_α, with

$$I_\alpha = \{\theta: f(\theta|\vec{x}) \le k\}$$

where k is chosen so that $\int_{I_\alpha} f(\theta|\vec{x})\, d\theta = \alpha$.

For multimodal densities $I_{1-\alpha}$ can be a set of disjoint intervals. The above approach is discussed by Berger (1985, p. 142). Also, Isaacs, Christ, Novick and Jackson (1974) produced tables giving I_α for standard density functions and various values of α. This subroutine HPD was first used in Pham-Gia and Turkkan (1990) to compute exact HPD intervals there. Due to the convergence of the posterior distribution toward the normal, approximate $(1 - \alpha)$-credible intervals can be computed, using either the normal distribution or, in some instances, Fieller's theorem (Hunter and Lamboy, 1981).

REMARKS:

a) Since the general idea of highest density can be aplied to any density $f(\theta)$, the interval $I_{1-\alpha}$ above is also known as highest probability interval. However, it is more meaningful to consider such an interval in a Bayesian context.
b) The algorithm is general and can handle unimodal, multimodal, increasing, decreasing or bathtub densities without any difficulty.

METHOD

For a given value of α and an input density $f(\theta)$, an $(1 - \alpha)$-HPD credible interval can be computed according to the following algorithm:

a) Start with a value k such that $k > f(\theta)$, $\forall\theta$.
b) Compute $I = \int_{\theta_L}^{\theta_R} f(\theta)\, d\theta$ while setting $f(\theta) = 0$ if $f(\theta) > k$. Here, θ_L, θ_R are the left and right bounds of the range of input density.
c) Check if $I = \alpha$. If not, bisect $(0,k)$ and return to b) until equality is obtained.
d) When $I = \alpha$, compute θ_i by solving $f(\theta_i) = k$, $i = 1, 2, \ldots$

Integrations are carried out by using an adaptive 8-point Gauss-Legendre quadrature, and the equation mentioned in d) is solved by the simple bisection technique.

STRUCTURE

SUBROUTINE HPD (TR,TL,ALFA,BIGK,QI,HPDI,IVAL,IFAULT)

Formal parameters

TL	Real	input	: θ_L
TR	Real	input	: θ_R
ALFA	Real	input	: α
BIGK	Real	input	: A high k value
QI	Real	output	: Value of I
HPDI	Real array ( )	output	: Contains end points of HPD interval
IVAL	Integer	output	: The number of HPD intervals

IFAULT	Integer	output : An error indicator IFAULT = 0, no errors IFAULT = 1, TL > TR IFAULT = 2, ALFA < 0 or ALFA > 0.6 IFAULT = 3, $f(\theta)$ is not a probability function or the initial value of k is too small.

AUXILIARY ALGORITHMS

Subroutines G8, QAG, BISECT and function PDF

The purpose of these subroutines and function is as follows:

G8 — 8-point Gauss-Legendre quadrature.
QAG — 16-level deep adaptive quadrature which uses G8.
BISECT — General purpose bisection routine. Computes the root of a function which is known to lie between the two bounds $\times 1$ and $\times 2$.
PDF — User-supplied probability function.

PRECISION AND TIMING

Testing of the algorithm was done in single precision on a VAX-6310 computer using different types of densities, and no difficulty was encountered. Usual densities considered include the increasing, decreasing, J-shaped, reversed J-shaped, unimodal and bathtub types (e.g., Gamma, Beta, Power, Log-normal distributions, etc.). Moreover, multimodal densities (e.g., a weighted average of several Beta's) are also considered and the resulting HPD regions are sets of disjoint intervals given as outputs. Using this routine intervals obtained are identical to those given by Isaacs *et al.* (1974), for several densities listed there. Timing checks are not reported here, since computing time was usually under a few seconds.

References

Abramowitz, M. and Stegun, I. A. (1972). *Handbook of Mathematical Functions with Formulas, Graphs and Mathematical Tables,* Dover: New York.

Berger, J. O. (1985). *Statistical Decision Theory and Bayesian Analysis,* 2nd ed. New York: Springer-Verlag.

Box. G. E. P. and Tiao, G. (1973). *Bayesian Inference in Statistical Analysis.* Reading, Massachusetts: Addison-Wesley.

Gerald, C. F., Wheatley, P. O. (1990). *Applied Numerical Analysis.* Reading, Massachusetts: Addison-Wesley.

Hunter, W. G. and Lamboy, W. F. (1981). A Bayesian Analysis of the Linear Calibration Problem (with discussions), *Technometrics*, **23**, 323–350.
Isaacs, G. L., Christ, D. E., Novick, M. R. and Jackson, P. H. (1974). *Tables for Bayesian Statisticians.* Iowa City: The University of Iowa Press.
Pham-Gia, T. and Turkkan, N. (1992). *Information Loss in a Noisy Dichotomous Sampling Process. Comm. in Stat, Theory and Methods.* (to appear in Sept. 1992 issue).
Press, W. H., Flannery, B. P., Teukolsky, S. A. and Vetterling, W. T. (1986). *Numerical Recipes.* Cambridge: Cambridge University Press.

```
      SUBROUTINE HPD (TL,TR,ALFA,BIGK,QI,HPDI,IVAL,IFAULT)
C
C----- Computes (1-alpha) HPD credible interval
C
      REAL TL,TR,ALFA,T(20),P,DK,KU,KL,T1,T2,
     +      EPS3,EPS7,K,BIGK,PF1,PF2,HPDI(10,2),
     +      FJT,FJT2,EPS2
      INTEGER JT,J,IFAULT,IVAL,KJ
      COMMON/HPD1/K
      PARAMETER (EPS2=1.0E-2,EPS3=1.0E-3,EPS7=1.0E-7)
C
C----- Check for errors
C
      IFAULT=0
      IF (TL.GE.TR) THEN
         IFAULT=1
         RETURN
      END IF
      IF (ALFA.LT.0.0E0 .OR. ALFA.GT.0.6E0) THEN
         IFAULT=2
         RETURN
      END IF
      K=BIGK
      CALL QAG (TL,TR,QI,EPS3)
      IF (ABS(1.0E0-QI).GT.EPS2) THEN
         IFAULT=3
         RETURN
      END IF
      IF (ALFA.EQ.0.0E0) THEN
         IVAL=1
         HPDI(1,1)=TL
         HPDI(1,2)=TR
         RETURN
      END IF
C
C----- Begin by 0 < k < bigk then
```

```
C----- bisect until I=alfa
C
        KU=K
        KL=0.0E0
10      CONTINUE
           CALL QAG (TL,TR,P,EPS3)
           IF (P.LT.ALFA) THEN
              KL=K
              K=0.5E0*(K+KU)
              DK=KL-K
           ELSE
              KU=K
              K=0.5E0*(K+KL)
              DK=KU-K
           END IF
           IF(ABS(P-ALFA).LE.EPS3*ALFA.OR.ABS(DK).LT.EPS7)
     +        GOTO 100
        GOTO 10
C
C----- k is found, sweep the interval (TL,TR)
C----- to locate abscissas where f(theta)-k=0
C
100     CONTINUE
        J=1
        T1=TL
15      CONTINUE
           T1=T1+EPS3
           T2=T1+EPS3
           IF (T2.GE.TR) GOTO 20
           PF1=PDF(T1)-K
           PF2=PDF(T2)-K
           IF((PF1*PF2).LE.0.0E0) THEN
              J=J+1
              CALL BISECT (T1,T2,T(J))
           END IF
        GOTO 15
20      CONTINUE
        JT=J+1
        T(1)=TL
        T(JT)=TR
        FJT=FLOAT(JT)
        FJT2=FJT-INT(FJT/2.0E0)*2.0E0
        IVAL=(JT-1)/2
        IF ((PDF(T(2)+EPS3)).GT.K) THEN
           IF(FJT2.EQ.0.0E0) IVAL=(JT-2)/2
           KJ=1
```

```
      DO 30 J=1,IVAL
         HPDI(J,1)=T(KJ+1)
         HPDI(J,2)=T(KJ+2)
         KJ=KJ+2
30    CONTINUE
   ELSE
      IF(FJT2.EQ.0.0E0) IVAL=JT/2
      KJ=1
      DO 40 J=1,IVAL
         HPDI(J,1)=T(KJ)
         HPDI(J,2)=T(KJ+1)
         KJ=KJ+2
40    CONTINUE
   END IF
   RETURN
   END
   SUBROUTINE G8 (A,B,S)
C
C------8-point Gauss-Legendre quadrature
C
   REAL XG(4),WG(4),A,B,XM,XR,S,DG,F1,F2,K
   INTEGER I
   COMMON/HPD1/K
   DATA XG /0.18343464250E0,0.52553240992E0,
  +          0.79666647741E0,0.96028985650E0/
   DATA WG /0.36268378338E0,0.31370664588E0,
  +          0.22238103445E0,0.10122853629E0/
   XM=0.5E0*(B+A)
   XR=0.5E0*(B-A)
   S=0.0E0
   DO 10 I=1,4
      DG=XR*XG(I)
      F1=PDF(XM+DG)
      F2=PDF(XM-DG)
      IF (F1.GT.K) F1=0.0E0
      IF (F2.GT.K) F2=0.0E0
      S=S+WG(I)*(F1+F2)
10 CONTINUE
   S=S*XR
   RETURN
   END
   SUBROUTINE QAG(A,B,ANS,TOL)
C
C------16-level deep adaptive quadrature using
C------8-point Gauss-Legendre algorithm.
C
```

```
      REAL S(4,16),A,B,ANS,TOL,AX,BX,H1,H2,C,
     +       S1AB,S2AB,S2AC,S2CB,TOLX,EPS7
      INTEGER L,MAXLV,LR(16)
      PARAMETER (MAXLV=16,EPS7=1.0E-7)
      TOLX=TOL
      AX=A
      H1=B-A
      L=1
      LR(L)=1
      S(4,L)=0.0E0
      CALL G8(A,B,S1AB)
10    CONTINUE
         H2=0.5E0*H1
         C=AX+H2
         BX=C+H2
         CALL G8 (AX,C,S2AC)
         CALL G8 (C,BX,S2CB)
         S2AB=S2AC+S2CB
         IF(ABS(S2AB-S1AB).LE.TOLX*ABS(S2AB)) GOTO 20
         IF(ABS(H2).LE.EPS7*ABS(AX)) GOTO 20
         IF(L.GE.MAXLV) GOTO 20
         L=L+1
         LR(L)=0
         H1=H2
         TOLX=0.5E0*TOLX
         S(1,L)=H1
         S1AB=S2AC
         S(2,L)=S2CB
         S(3,L)=TOLX
      GOTO 10
40    CONTINUE
         S(4,L)=S2AB
         LR(L)=1
         AX=AX+H1
         H1=S(1,L)
         S1AB=S(2,L)
         TOLX=S(3,L)
      GOTO 10
20    CONTINUE
      IF(LR(L).EQ.0) GOTO 40
      S2AB=S2AB+S(4,L)
      L=L-1
      IF(L.GT.1) GOTO 20
      ANS=S2AB
```

```
      RETURN
      END
      SUBROUTINE BISECT (X1,X2,X)
C
C------General purpose bisection routine
C------Computes the root of f(x)-k=0
C------between ×1 and ×2
C
      REAL X1,X2,X,K,EPS7,FX,DX,FMID,XMID
      COMMON/HPD1/K
      PARAMETER (EPS7=1.0E-7)
      FX=PDF(X1)-K
      IF (FX.LT.0.0E0) THEN
        X=X1
        DX=X2-X1
      ELSE
          X=X2
          DX=X1-X2
      END IF
10    CONTINUE
          DX=0.5E0*DX
          XMID=X+DX
          FMID=PDF(XMID)-K
          IF (FMID.LT.0.0E0) X=XMID
          IF (ABS(DX).LT.EPS7.OR.FMID.EQ.0.0E0) RETURN
      GOTO 10
      END
```